

Tema 11

Funcions inverses. Funcions implícites. Fórmula de Taylor. Extrems.

Objectius:

1. Practicar el Teorema de la Funció Inversa.
2. Practicar el Teorema de la Funció Implícita.
3. Practicar el càlcul de derivades de funcions definides implícitament.
4. Calcular el polinomi de Taylor de funcions de diverses variables.
5. Estudiar els extrems (locals, condicionats i absoluts) de funcions de diverses variables.

Continguts:

- 11-1. El teorema de la funció inversa. Aplicacions.
- 11-2. El teorema de la funció implícita. Derivació de funcions implícites.
- 11-3. Aproximació de funcions diferenciables. Fórmula de Taylor.
- 11-4. Extrems de funcions reals de diverses variables. Extrems locals, extrems condicionats i extrems absoluts.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.
- AP86 APOSTOL, T.M. (1986)
Análisis Matemático

- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima
- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

11-1.- El teorema de la funció inversa. Aplicacions.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_11-1.wxm**.

11.1.1.- El teorema de la funció inversa.

Inversa local d'una funció. Si una funció $f : A \rightarrow B$ és bijectiva, és a dir, per a cada $y \in B$ existeix un únic $x \in A$ tal que $f(x) = y$, aleshores existeix la funció inversa de f , $f^{-1} : B \rightarrow A$ definida per

$$f^{-1}(y) = x \Leftrightarrow f(x) = y$$

Hi ha molts casos en que una funció no és bijectiva, i per tant no existeix la seva funció inversa, però tanmateix sí que ho és en l'entorn d'un punt, en el sentit que precisa la definició següent. Sigui $f : A \subset \mathbb{R}^n \rightarrow \mathbb{R}^n$ una funció; es diu que aquesta funció és localment bijectiva en un punt $a \in A$ en el qual $f(a) = b \in B$ si existeix un entorn $U(a)$ del punt $a \in A$ tal que si $V(b) = f(U(a))$, la funció $f : U(a) \rightarrow V(b)$ és bijectiva. Aleshores, s'anomena inversa local de f , la funció definida per

$$\begin{aligned} f^{-1} : V(b) &\rightarrow U(a) \\ y &\rightarrow x = f^{-1}(y) \Leftrightarrow f(x) = y \end{aligned}$$

El resultat general és el que estableix el teorema que s'enuncia a continuació.

Teorema (de la funció inversa). Sigui una funció $f : A \subset \mathbb{R}^n \rightarrow \mathbb{R}^n$, on A és un conjunt obert i $f \in C^1(A)$; si es considera un punt $a \in A$ tal que $\det Jf(a) \neq 0$, aleshores existeixen dos conjunts oberts $U(a), V(b)$ entorns respectius dels punts $a \in A$ i $b = f(a)$, tals que:

- 1) la funció $f : U(a) \rightarrow V(b)$ és bijectiva;
- 2) la funció inversa compleix $f^{-1} \in C^1(V(b))$;
- 3) Per a cada punt $y = f(x) \in V(b)$ es compleix $Jf^{-1}(y) = (Jf(x))^{-1}$.

Exemple 11.1.1. Considerem la funció $f : \mathbb{R}^2 \rightarrow \mathbb{R}^2$ definida per

$$f(x, y) = (e^x, e^{x+y}), (x, y) \in \mathbb{R}^2.$$

De forma equivalent, la funció es pot definir mitjançant el sistema d'equacions:

$$[1] \quad \begin{cases} u = f_1(x, y) = e^x \\ v = f_2(x, y) = e^{x+y} \end{cases}$$

Vegem aquestes definicions en wxMaxima:

```
(%i1) f1(x,y):=exp(x);      f2(x,y):=exp(x+y);
      f12(x,y):=[f1(x,y), f2(x,y)];
(%o1) f1(x,y):=exp(x)
(%o2) f2(x,y):=exp(x+y)
(%o3) f12(x,y):=[f1(x,y), f2(x,y)]

(%i4) Eq11:u=f1(x,y);      Eq12:v=f2(x,y);
(%o4) u = %e^x
(%o5) v = %e^y+x
```

La funció considerada és diferenciable en qualsevol punt del seu domini $a = (x, y)$. Calculem la matriu jacobiana de la funció en aquest punt: per fer-ho disposem, com és sabut, d'una instrucció específica de wxMaxima

```
(%i6) JF(x,y)=Jfxy: jacobian ( [f1(x,y), f2(x,y)] , [x,y] );
(%o6) JF(x,y)=
$$\begin{bmatrix} \%e^x & 0 \\ \%e^{y+x} & \%e^{y+x} \end{bmatrix}$$

```

Considerem, per exemple, el punt $a = (0, 0)$, que compleix $f(0, 0) = (1, 1)$. Per calcular la matriu jacobiana en aquest punt cal definir les funcions derivades parcials:

```
(%i20) Dxf1(x,y):=%e^x$      Dyf1(x,y):=0$
      Dxf2(x,y):=%e^(x+y)$    Dyf2(x,y):=%e^(x+y)$
(%i24) Jf(0,0)=Jf00:matrix( [Dxf1(0,0), Dyf1(0,0)] , [Dxf2(0,0), Dyf2(0,0)]);
(%o24) Jf(0,0)=
$$\begin{bmatrix} 1 & 0 \\ 1 & 1 \end{bmatrix}$$

```

Ara calculem el jacobí, és a dir, el determinant de la matriu jacobiana en aquest punt:

```
(%i8) '(detJf(0,0))=determinant(Jf00);
(%o8) detJf(0,0)=1
```

Atès que el jacobí és no nul, la funció és localment invertible en un entorn del punt. L'aproximació lineal de la funció en un entorn d'aquest punt és

$$f(x, y) = f(a) + Jf(a)h$$

Per tant:

```
(%i13) Mxy=Mxy:matrix([x],[y]);
(%o13) Mxy= $\begin{bmatrix} x \\ y \end{bmatrix}$ 

(%i14) 'f(0,0)=f00:matrix( [f1(0,0)] , [f2(0,0)] ) ;
(%o14) f(0,0)= $\begin{bmatrix} 1 \\ 1 \end{bmatrix}$ 

(%i15) Muv=Muv:matrix([u],[v]);
(%o15) Muv= $\begin{bmatrix} u \\ v \end{bmatrix}$ 

(%i16) Muv = f00 + Jf00.Mxy;
(%o16)  $\begin{bmatrix} u \\ v \end{bmatrix} = \begin{bmatrix} x+1 \\ y+x+1 \end{bmatrix}$ 
```

és a dir, l'aproximació lineal del sistema [1] en un entorn del punt $a = (0,0)$ és:

```
(%i17) Eq11_lin:u=1+x;
      Eq12_lin:v=1+x+y;
(%o17) u=x+1
(%o18) v=y+x+1
```

Ara calcularem l'aproximació lineal de la funció inversa aplicant el Teorema de la funció inversa. Per això calcularem la inversa de la matriu jacobiana:

```
(%i19) 'Jf_inv(1,1)=inv_Jf00:invert(Jf00);
(%o19) Jf_inv(1,1)= $\begin{bmatrix} 1 & 0 \\ -1 & 1 \end{bmatrix}$ 
```

I finalment aplicarem l'equació:

$$f^{-1}(u, v) = f^{-1}(f(a)) + Jf^{-1}(f(a))h$$

Els càlculs amb wxMaxima són:

```
(%i19) 'Jf_inv(1,1)=inv_Jf00:invert(Jf00);
(%o19) Jf_inv(1,1)= $\begin{bmatrix} 1 & 0 \\ -1 & 1 \end{bmatrix}$ 

(%i20) 'h=h:matrix([u-1],[v-1]);
(%o20) h= $\begin{bmatrix} u-1 \\ v-1 \end{bmatrix}$ 

(%i21) f_inv(1,1)=finv11: matrix( [0], [0] );
(%o21) f_inv(1,1)= $\begin{bmatrix} 0 \\ 0 \end{bmatrix}$ 

(%i22) Mxy = finv11 + inv_Jf00.h;
(%o22)  $\begin{bmatrix} x \\ y \end{bmatrix} = \begin{bmatrix} u-1 \\ v-u \end{bmatrix}$ 
```

és a dir, l'aproximació lineal de la funció inversa en un entorn del punt (1,1) és:

```
(%i23) Eq21_lin:x=u-1;
      Eq22_lin:y=v-u;
(%o23) x=u-1
(%o24) y=v-u
```

11.1.2 Sistemes de coordenades. canvi de variables

Definició. Un sistema de coordenades de tipus C^r en un obert $U \subset \mathbb{R}^n$ és una parella (ϕ, W) en la qual

- W és un obert de $\mathbb{R}^n$
- $\phi: W \rightarrow U$ és bijectiva
- $\phi \in C^r(W)$ i $\phi^{-1} \in C^r(U)$

Si en un punt $x = (x_1, x_2, \dots, x_n) \in U$ es compleix $\phi(\alpha) = x$ amb $\alpha = (\alpha_1, \alpha_2, \dots, \alpha_n) \in W$, els nombres reals $\alpha_1, \alpha_2, \dots, \alpha_n$ s'anomenen coordenades de x en el sistema de coordenades (ϕ, W) . La matriu $J\phi(\alpha)$ s'anomena jacobiana del sistema de coordenades i el seu determinant, jacobià del sistema de coordenades.

Exemple 11.1.2.

(a) Coordenades polars planes. Si $U = \mathbb{R}^2 - \{0\}$, la parella (ϕ, W) en la qual l'aplicació $\phi: W \rightarrow U$ està definida per

$$\phi(r, \theta) = (r \cos \theta, r \sin \theta), (r, \theta) \in W$$

i $W =]0, +\infty[\times]-\pi, \pi[$ és un sistema de coordenades en U anomenat coordenades polars planes. Calculem la matriu jacobiana i el determinant jacobí d'aquest sistema de coordenades amb wxMaxima:

```
(%i1) f1(r, theta):=r*cos(theta);    f2(r, theta):=r*sin(theta);
      f(r,theta):=[f1(r, theta), f2(r, theta)];
(%o1) f1(r, theta):=r*cos(theta)
      (%o2) f2(r, theta):=r*sin(theta)
      (%o3) f(r, theta):=[f1(r, theta), f2(r, theta)]

(%i4) Jf(r, theta)=Jfcpp: jacobian ( [f1(r,theta), f2(r,theta)] , [r,theta] );
(%o4) Jf(r, theta)=
      [cos(theta)  -r sin(theta)]
      [sin(theta)   r cos(theta)]

(%i7) '(detJf(r,theta))=determinant(Jfcpp);
(%o7) detJf(r, theta)=r sin(theta)^2 + r cos(theta)^2

(%i8) trigsimp(%);
(%o8) detJf(r, theta)=r
```

(b) Coordenades cilíndriques. Si $U = \mathbb{R}^3 - \{0\}$, la parella (ϕ, W) en la qual l'aplicació $\phi: W \rightarrow U$ està definida per

$$\phi(r, \theta, z) = (r \cos \theta, r \sin \theta, z), (r, \theta, z) \in W$$

i $W =]0, +\infty[\times]-\pi, \pi[\times \mathbb{R}$ és un sistema de coordenades en U anomenat coordenades cilíndriques. Calculem la matriu jacobiana i el determinant jacobí d'aquest sistema de coordenades amb wxMaxima:

```
(%i1) f1(r, theta, z):=r*cos(theta); f2(r, theta, z):=r*sin(theta); f3(r, theta, z):=z;
      f(r,theta,z):=[f1(r, theta), f2(r, theta), f3(r, theta)];
(%o1) f1(r, theta, z):=r*cos(theta)
      (%o2) f2(r, theta, z):=r*sin(theta)
      (%o3) f3(r, theta, z):=z
      (%o4) f(r, theta, z):=[f1(r, theta), f2(r, theta), f3(r, theta)]
```

```
(%i5) Jf(r, theta, z)=Jfcc: jacobian
      ([f1(r, theta, z), f2(r, theta, z), f3(r, theta, z)], [r, theta, z]);

(%o5) Jf(r, theta, z) = 
$$\begin{bmatrix} \cos(\theta) & -r \sin(\theta) & 0 \\ \sin(\theta) & r \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$


(%i6) '(detJf(r, theta))=determinant(Jfcc);
(%o6) detJf(r, theta)=r sin(theta)^2 + r cos(theta)^2

(%i7) trigsimp(%);
(%o7) detJf(r, theta)=r
```

(c) Coordenades esfèriques. Si $U = \mathbb{R}^3 - \{0\}$, la parella (ϕ, W) en la qual l'aplicació $\phi: W \rightarrow U$ està definida per

$$\phi(r, \varphi, \theta) = (r \cos \varphi \cos \theta, r \cos \varphi \sin \theta, r \sin \varphi), (r, \varphi, \theta) \in W$$

i $W =]0, +\infty[\times]-\frac{\pi}{2}, \frac{\pi}{2}[\times]-\pi, \pi[$ és un sistema de coordenades en U anomenat coordenades esfèriques. Calculem la matriu jacobiana i el determinant jacobí d'aquest sistema de coordenades amb wxMaxima:

```
(%i1) f1(r, phi, theta):=r*cos(phi)*cos(theta);
      f2(r, phi, theta):=r*cos(phi)*sin(theta);
      f3(r, phi, theta):=r*sin(phi);
      f(r, phi, theta):=[f1(r, phi, theta), f2(r, phi, theta), f3(r, phi, theta)];

(%o1) f1(r, phi, theta):=r cos(phi) cos(theta)
(%o2) f2(r, phi, theta):=r cos(phi) sin(theta)
(%o3) f3(r, phi, theta):=r sin(phi)
(%o4) f(r, phi, theta):=[f1(r, phi, theta), f2(r, phi, theta), f3(r, phi, theta)]

(%i5) Jf(r, phi, theta)=Jfce: jacobian
      ([f1(r, phi, theta), f2(r, phi, theta), f3(r, phi, theta)], [r, phi, theta]);

(%o5) Jf(r, phi, theta) = 
$$\begin{bmatrix} \cos(\phi) \cos(\theta) & -\sin(\phi) r \cos(\theta) & -\cos(\phi) r \sin(\theta) \\ \cos(\phi) \sin(\theta) & -\sin(\phi) r \sin(\theta) & \cos(\phi) r \cos(\theta) \\ \sin(\phi) & \cos(\phi) r & 0 \end{bmatrix}$$

```



```
(%i6) '(detJf(r,theta))=determinant(Jfce);
(%o6) detJf(r,theta)=-cos(phi)r sin(theta)(sin(phi)^2 r sin(theta)+cos(phi)^2 r sin(theta))-cos(phi) sin(phi)^2 r^2 cos(theta)^2-cos(phi)^3 r^2
cos(theta)^2
(%i7) trigsimp(%);
(%o7) detJf(r,theta)=-cos(phi)r^2
```

Els sistemes de coordenades s'apliquen per transformar algunes equacions i obtenir una representació que permeti una major facilitat per resoldre-les. Aquesta metodologia es fonamenta en la definició següent: si $f: A \subset \mathbb{R}^n \rightarrow \mathbb{R}^m$ i (ϕ, W) és un sistema de coordenades definit en A , la funció composta $\tilde{f} = f \circ \phi$ s'anomena expressió de la funció en el sistema de coordenades considerat. Quan una funció s'expressa en un sistema de coordenades diferent de les coordenades cartesianes, es diu que s'ha fet un canvi de variables.

Exemple 11.1.3. Considerem la funció $f(x, y) = x^2 + 3y^2$, $(x, y) \in \mathbb{R}^2$. Es vol obtenir l'expressió d'aquesta funció en coordenades polars planes, és a dir, fer un canvi de variables a coordenades polars planes. Els càlculs amb wxMaxima permeten obtenir fàcilment l'expressió de la funció en aquestes coordenades:

```
(%i1) f(x,y):=x^2 + 3*y^2;
(%o1) f(x,y):=x^2 + 3 y^2

(%i2) x:=r*cos(theta)$ y:=r*sin(theta)$
      F(r,theta)=f(x,y);
(%o4) F(r,theta)=3 r^2 sin(theta)^2 + r^2 cos(theta)^2

(%i5) trigsimp(%);
(%o5) F(r,theta)=2 r^2 sin(theta)^2 + r^2

(%i6) factor(%);
(%o6) F(r,theta)=r^2 (2 sin(theta)^2 + 1)
```

Aquesta és l'expressió de la funció en coordenades polars planes, que té l'avantatge de la separació de les variables en producte de dos factors d'una variable.

El canvi de variables és útil no només en l'expressió de funcions, sinó també en la transformació d'equacions en les que intervenen funcions i les seves derivades (equacions diferencials), tal com es pot veure a l'exemple següent.

Exemple 11.1.4. Es considera l'equació

$$[1] \quad \left(x + 2y\sqrt{x^2 + y^2}\right) D_x f(x, y) + \left(y - 2x\sqrt{x^2 + y^2}\right) D_y f(x, y) = 0$$

Es tracta de calcular l'equació que compleix l'expressió de la funció en coordenades polars planes (és a dir, transformar l'equació mitjançant el canvi de variables a coordenades polars planes).

En primer lloc expressem les coordenades polars en funció de les cartesianes:

```
(%i1) r(x,y):=+sqrt(x^2 + y^2);
      th(x,y):=atan(y/x);
(%o1) r(x,y):=sqrt(x^2 + y^2)
(%o2) th(x,y):=atan(y/x)
```

Observeu que cal canviar la denominació de “theta” per no interferir en càlculs posteriors. Si es designa per F l'expressió de la funció f en coordenades polars planes, per definició es compleix $F = f \circ \phi$; per tant, $f = F \circ \phi^{-1} = F(r, \theta)$. A partir d'aquesta expressió, es tracta ara de calcular l'expressió de les derivades parcials de la funció f . Aplicant la regla de la cadena resulta:

$$\begin{aligned} D_x f(x, y) &= D_r F(r, \theta) D_x r + D_\theta F(r, \theta) D_x \theta \\ D_y f(x, y) &= D_r F(r, \theta) D_y r + D_\theta F(r, \theta) D_y \theta \end{aligned}$$

Fem els càlculs amb wxMaxima. Calculem les derivades parcials de $r(x, y)$:

```
(%i3) diff(r(x,y), x);    diff(r(x,y), y);
(%o3) x/sqrt(y^2+x^2)
(%o4) y/sqrt(y^2+x^2)
```

Ara les expressem com funcions:

```
(%i5) Dxr(x,y):=x/sqrt(y^2+x^2);    Dyr(x,y):=y/sqrt(y^2+x^2);
(%o5) Dxr(x,y):=x/sqrt(y^2+x^2)
(%o6) Dyr(x,y):=y/sqrt(y^2+x^2)
```

Calculem les derivades parcials de $\theta(x, y)$:

```
(%i7) ratsimp(diff(th(x,y), x)); ratsimp(diff(th(x,y), y));
```

```
(%o7) 
$$-\frac{y}{y^2+x^2}$$

```

```
(%o8) 
$$\frac{x}{y^2+x^2}$$

```

Ara les expressem com funcions:

```
(%i9) Dxth(x,y):=-y/(y^2+x^2); Dyth(x,y):=x/(y^2+x^2);
```

```
(%o9) 
$$Dxth(x,y):=-\frac{y}{y^2+x^2}$$

```

```
(%o10) 
$$Dyth(x,y):=\frac{x}{y^2+x^2}$$

```

Substituint a la primera de les equacions s'obté:

```
(%i11) Dxf=Dxf:DrF(r,theta)*Dxr(x,y) + DthF(r,theta)*Dxth(x,y);
```

```
(%o11) 
$$Dxf = \frac{DrF(r,\theta) x}{\sqrt{y^2+x^2}} - \frac{DthF(r,\theta) y}{y^2+x^2}$$

```

Substituint a la segona de les equacions s'obté:

```
(%i12) Dyf=Dyf:DrF(r,theta)*Dyr(x,y) + DthF(r,theta)*Dyth(x,y);
```

```
(%o12) 
$$Dyf = \frac{DrF(r,\theta) y}{\sqrt{y^2+x^2}} + \frac{DthF(r,\theta) x}{y^2+x^2}$$

```

Calculem ara el primer terme de l'equació [1]:

```
(%i13) T1:(x + 2*y*sqrt(x^2+y^2))*Dxf;
```

```
(%o13) 
$$\left( \frac{DrF(r,\theta) x}{\sqrt{y^2+x^2}} - \frac{DthF(r,\theta) y}{y^2+x^2} \right) (2 y \sqrt{y^2+x^2} + x)$$

```

```
(%i14) T1:expand(%);
```

```
(%o14) 
$$-\frac{2 DthF(r,\theta) y^2}{\sqrt{y^2+x^2}} + \frac{DrF(r,\theta) x^2}{\sqrt{y^2+x^2}} - \frac{DthF(r,\theta) x y}{y^2+x^2} + 2 DrF(r,\theta) x y$$

```

Calculem ara el segon terme de l'equació [1]:

```
(%i15) T2:(y - 2*x*sqrt(x^2+y^2))*Dyf;
```

```
(%o15) 
$$\left( \frac{DrF(r,\theta) y}{\sqrt{y^2+x^2}} + \frac{DthF(r,\theta) x}{y^2+x^2} \right) (y - 2 x \sqrt{y^2+x^2})$$

```

```
(%i16) T2:expand(%);
```

```
(%o16) 
$$\frac{\text{DrF}(r, \theta) y^2}{\sqrt{y^2 + x^2}} - \frac{2 \text{DthF}(r, \theta) x^2}{\sqrt{y^2 + x^2}} + \frac{\text{DthF}(r, \theta) x y}{y^2 + x^2} - 2 \text{DrF}(r, \theta) x y$$

```

Substituint i igualant a zero tindrem la nova forma de l'equació [1]:

```
(%i17) T1+T2=0;
```

```
(%o17) 
$$-\frac{2 \text{DthF}(r, \theta) y^2}{\sqrt{y^2 + x^2}} + \frac{\text{DrF}(r, \theta) y^2}{\sqrt{y^2 + x^2}} - \frac{2 \text{DthF}(r, \theta) x^2}{\sqrt{y^2 + x^2}} + \frac{\text{DrF}(r, \theta) x^2}{\sqrt{y^2 + x^2}} = 0$$

```

Operant i simplificant, s'obté:

```
(%i18) factor(%);
```

```
(%o18) 
$$-(2 \text{DthF}(r, \theta) - \text{DrF}(r, \theta)) \sqrt{y^2 + x^2} = 0$$

```

Finalment, l'expressió buscada és:

```
(%i19) %*(1/sqrt(x^2+y^2));
```

```
(%o19) 
$$\text{DrF}(r, \theta) - 2 \text{DthF}(r, \theta) = 0$$

```

expressió que, com resulta evident, és molt més senzilla que [1].

11-2.- El teorema de la funció implícita. Derivació de funcions implícites.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_11-2.wxm**.

11.2.1 El teorema de la funció implícita

Teorema. Sigui $f: A \times B \subset \mathbb{R}^{n+p} \rightarrow \mathbb{R}^p$ una funció i sigui $(a,b) \in A \times B$. Si es compleixen les condicions:

- f és de tipus C^r en $A \times B$
- $f(a,b) = 0$
- $\det(D_{n+k} f_j(a,b))_{\substack{1 \leq k \leq p \\ 1 \leq j \leq p}} \neq 0$

aleshores existeixen $W \subset \mathbb{R}^n$ entorn obert del punt $a \in A$ i una funció $\varphi: W \rightarrow \mathbb{R}^p$ tals que:

- 1) φ és de tipus C^r en W ;
- 2) $\varphi(a) = b$;
- 3) $f(x, \varphi(x)) = 0$ per a tot $x \in W$.

La funció φ es diu definida implícitament pel sistema

$$\begin{cases} f_1(x_1, x_2, \dots, x_{n+p}) = 0 \\ \dots \\ f_p(x_1, x_2, \dots, x_{n+p}) = 0 \end{cases}$$

Exemple 11.2.1. Es considera l'equació

$$x + y + z + \cos(xyz) = 0$$

Es tracta de veure que aquesta equació defineix una funció implícita $z = z(x, y)$ en un entorn del punt $a = (0, 0, -1)$. Per fer-ho, caldrà verificar que es compleixen les condicions establertes al Teorema de la funció implícita. En primer lloc es pot afirmar que la funció

$$f(x, y, z) = x + y + z + \cos(xyz)$$

és de tipus C^∞ en $\mathbb{R}^3$. Definim la funció amb wxMaxima i calculem el seu valor en el punt considerat:

```
(%i2) f(x,y,z):=x+y+z+cos(x*y*z); 'f(0,0,-1)=f(0,0,-1);
(%o2) f(x,y,z):=x+y+z+cos(x*y*z)
(%o3) f(0,0,-1)=0
```

A continuació calcularem la derivada de la funció respecte de la variable que es demana verificar que es pot expressar com funció implícita:

```
(%i5) Dzf(x,y,z)=diff(f(x,y,z), z);
(%o5) Dzf(x,y,z)=1-x*y*sin(x*y*z)
```

Definim aquesta derivada parcial com funció i calculem el seu valor al punt:

```
(%i6) Dzf(x,y,z):=1-x*y*sin(x*y*z)$
'Dzf(0,0,-1)=Dzf(0,0,-1);
(%o7) Dzf(0,0,-1)=1
```

Per tant es compleixen les condicions del Teorema i, en efecte, l'equació donada defineix una funció implícita $z = z(x, y)$ en un entorn del punt $a = (0, 0, -1)$. Es pot indicar al programa que a partir d'aquest moment la variable z depèn de x, y , cosa que es fa amb la instrucció:

```
(%i7) depends([z], [x,y]);
(%o7) [z(x,y)]
```

Finalment, l'equació inicial es pot escriure:

```
(%i8) Eq1b : x+y+z(x,y)+cos(x*y*z(x,y))=0;
(%o8) cos(x*z(x,y)*y)+y+z(x,y)+x=0
```

Exemple 11.2.2. Es considera el sistema d'equacions

$$\begin{cases} x^2 + y^2 - xz^2 - 1 = 0 \\ xz - x^2y = 0 \end{cases}$$

Es tracta d'estudiar si en un entorn del punt $P = (x_0, y_0, z_0) = (1, 1, 1)$:

- 1) el sistema defineix x, y com a funcions implícites de z ;
- 2) el sistema defineix x, z com a funcions implícites de y ;
- 3) el sistema defineix y, z com a funcions implícites de x .

Caldrà veure en cada cas si es compleixen les condicions del Teorema de la funció implícita. La primera es compleix: la diferenciabilitat de les funcions que defineixen cada una de les equacions.

Ara definirem en wxMaxima les equacions, les funcions que defineixen cada una de les equacions i verifiquem que el punt indicat compleix el sistema:

```
(%i1) Eq1: x^2 + y^2 - x*z^2 - 1 = 0;
      Eq2: x*z - x^2*y=0;
(%o1) -x z^2 + y^2 + x^2 - 1 = 0
      (%o2) x z - x^2 y = 0

(%i3) f1(x,y,z):=x^2 + y^2 - x*z^2 - 1;      f2(x,y,z):=x*z - x^2*y;
      f(x,y,z):= [f1(x,y,z), f2(x,y,z)];
(%o3) f1(x,y,z):=x^2 + y^2 + (-x) z^2 - 1
      (%o4) f2(x,y,z):=x z - x^2 y
      (%o5) f(x,y,z):= [f1(x,y,z), f2(x,y,z)]

(%i6) 'f(1,1,1)=f(1,1,1);
(%o6) f(1,1,1)= [0,0]
```

A continuació ens ocupem de la tercera condició en cada un dels casos. Prèviament calcularem la matriu jacobiana de la funció vectorial que té per components les funcions que defineixen cada una de les equacions:

```
(%i7) Jf(x,y,z)=Jf:jacobian( [f1(x,y,z),f2(x,y,z)] , [x,y,z] );
(%o7) Jf(x,y,z)=

$$\begin{bmatrix} 2x - z^2 & 2y & -2xz \\ z - 2xy & -x^2 & x \end{bmatrix}$$

```

Cas 1. Calcularem la submatriu corresponent a les dues primeres variables:

```
(%i8) Jf12=Jf12:submatrix(Jf, 3);
(%o8) Jf12=

$$\begin{bmatrix} 2x - z^2 & 2y \\ z - 2xy & -x^2 \end{bmatrix}$$

```

Avaluarem aquesta matriu en el punt $P=(1,1,1)$; per fer-ho aplicarem la instrucció “subst([x_i=x_{i0}],M)” que permet substituir variables per valors numèrics en una matriu M. Concretament la sintaxi és:

```
(%i9) Jf12_a=Jf12_a:subst( [x=1,y=1,z=1] , Jf12);
(%o9) Jf12_a=

$$\begin{bmatrix} 1 & 2 \\ -1 & -1 \end{bmatrix}$$

```

Ara calcularem el determinant d'aquesta matriu:

```
(%i10) '(detJf12_a)=determinant(Jf12_a);
(%o10) detJf12_a = 1
```

Atès que és no nul es pot afirmar que el sistema defineix x, y com a funcions implícites de z .

Cas 2. Calcularem la submatriu corresponent a les variables primera i tercera:

```
(%i11) Jf13=Jf13:submatrix(Jf, 2);
(%o11) Jf13=
$$\begin{bmatrix} 2x-z^2 & -2xz \\ z-2xy & x \end{bmatrix}$$

```

Avaluarem aquesta matriu en el punt $P = (1,1,1)$:

```
(%i12) Jf13_a=Jf13_a:subst( [x=1,y=1,z=1] , Jf13);
(%o12) Jf13_a=
$$\begin{bmatrix} 1 & -2 \\ -1 & 1 \end{bmatrix}$$

```

I calcularem el determinant d'aquesta matriu:

```
(%i13) '(detJf13_a)=determinant(Jf13_a);
(%o13) detJf13_a = -1
```

Atès que és no nul es pot afirmar que el sistema defineix x, z com a funcions implícites de y .

Cas 3. Calcularem la submatriu corresponent a les variables segona i tercera:

```
(%i14) Jf23=Jf23:submatrix(Jf, 1);
(%o14) Jf23=
$$\begin{bmatrix} 2y & -2xz \\ -x^2 & x \end{bmatrix}$$

```

Avaluarem aquesta matriu en el punt $P = (1,1,1)$:

```
(%i15) Jf23_a=Jf23_a:subst( [x=1,y=1,z=1] , Jf23);
(%o15) Jf23_a=
$$\begin{bmatrix} 2 & -2 \\ -1 & 1 \end{bmatrix}$$

```

Calculem el determinant d'aquesta matriu:


```
(%i16) '(detJf23_a)=determinant(Jf23_a);
(%o16) detJf23_a = 0
```

Atès que és nul es pot afirmar que el sistema no defineix y, z com a funcions implícites de x .

11.2.2 Derivació de funcions implícites

El Teorema de la funció implícita permet afirmar que quan un sistema defineix una funció implícita, aquesta funció és de tipus C'' , és a dir, es poden calcular les derivades parcials d'aquesta funció. La metodologia per fer-ho es basa en el fet següent: quan un sistema defineix una funció implícita, les components d'aquesta són les variables que es poden expressar en funció de la resta; quan s'ha establert aquest fet, aquestes variables s'han de considerar com a components d'una funció (la funció definida implícitament) i la resta segueixen tenint la consideració de variables. Il·lustrem aquesta metodologia amb la continuació dels dos exemples anteriors.

Exemple 11.2.3. Se sap que l'equació

$$x + y + z + \cos(xyz) = 0$$

defineix una funció implícita $z = z(x, y)$ en un entorn del punt $a = (0, 0, -1)$ tal que $z(0, 0) = -1$. Es tracta de calcular el valor de les derivades parcials d'aquesta funció en el punt $(0, 0)$, és a dir, $D_x z(0, 0), D_y z(0, 0)$.

En primer lloc escriurem l'equació i li assignarem una referència:

```
(%i1) Eq1 : x+y+z+cos(x*y*z)=0;
(%o1) cos(x y z)+z+y+x = 0
```

Ara declararem que z és funció de x, y :

```
(%i2) depends([z], [x,y]);
(%o2) [z(x, y)]
```

A continuació es calcularà la derivada parcial de l'equació respecte a cada una de les variables x, y assumint que wxMaxima tindrà en compte la dependència assignada. En efecte:

```
(%i3) DxEq1 : diff(x+y+z+cos(x*y*z), x) = 0;
      DyEq1 : diff(x+y+z+cos(x*y*z), y) = 0;
(%o3)  $-\left(x y \left(\frac{d}{d x} z\right) + y z\right) \sin(x y z) + \frac{d}{d x} z + 1 = 0$ 
(%o4)  $-\left(x y \left(\frac{d}{d y} z\right) + x z\right) \sin(x y z) + \frac{d}{d y} z + 1 = 0$ 
```

Per avaluar les derivades parcials de la funció implícita se substituirà el valor de les variables en el punt:

```
(%i5) DxEq1_P : -(0*Dxz(0,0)*0+(-1)*0)*sin(0)+Dxz(0,0)+1=0;
      DyEq1_P : -(0*Dyz(0,0)*0+(-1)*0)*sin(0)+Dyz(0,0)+1=0;
(%o5)  $Dxz(0,0) + 1 = 0$ 
(%o6)  $Dyz(0,0) + 1 = 0$ 
```

Per tant:

```
(%i7) Dxz(0,0)=-1; Dyz(0,0)=-1;
(%o7)  $Dxz(0,0) = -1$ 
(%o8)  $Dyz(0,0) = -1$ 
```

Exemple 11.2.4. Se sap que el sistema d'equacions

$$[1] \quad \begin{cases} x^2 + y^2 - xz^2 - 1 = 0 \\ xz - x^2y = 0 \end{cases}$$

defineix dues funcions implícites $x = x(z)$, $y = y(z)$ en un entorn del punt $P = (1, 1, 1)$ tals que $x(1) = 1$, $y(1) = 1$. Es tracta de calcular les derivades de primer ordre i de segon ordre d'aquestes funcions, és a dir, $Dx(1)$, $Dy(1)$, $D^2x(1)$, $D^2y(1)$.

En primer lloc escriurem les equacions del sistema i les hi assignarem una referència:

```
(%i1) Eq1: x^2 + y^2 - x*z^2 - 1 = 0;
      Eq2: x*z - x^2*y=0;
(%o1)  $-x z^2 + y^2 + x^2 - 1 = 0$ 
(%o2)  $x z - x^2 y = 0$ 
```

Ara declararem que les variables x, y són funció de z :

```
(%i3) depends( [x,y] , [z]);
(%o3) [x(z),y(z)]
```

A continuació es calcula la derivada de cada una de les equacions respecte de la variable z tenint en compte la dependència assignada:

```
(%i4) DzEq1 : diff(x^2 + y^2 - x*z^2 - 1, z) = 0;
      DzEq2 : diff(x*z - x^2*y, z) = 0;
(%o4)  $-\left(\frac{d}{dz}x\right)z^2 - 2xz + 2y\left(\frac{d}{dz}y\right) + 2x\left(\frac{d}{dz}x\right) = 0$ 
(%o5)  $\left(\frac{d}{dz}x\right)z - x^2\left(\frac{d}{dz}y\right) - 2x\left(\frac{d}{dz}x\right)y + x = 0$ 
```

Per calcular les derivades $Dx(1)$, $Dy(1)$ cal aplicar els valors numèrics corresponents en aquestes equacions:

```
(%i6) DzEq1_P : - 1^2 * Dx(1) - 2*1*1 + 2*1*Dy(1) + 2*1*Dx(1) = 0;
      DzEq2_P : 1*Dx(1) - 1^2*Dy(1) - 2*1*Dx(1)*1 + 1 = 0;
(%o6) 2 Dy(1)+Dx(1)-2=0
(%o7) -Dy(1)-Dx(1)+1=0
```

Resolent aquest sistema lineal resulta:

```
(%i8) linsolve([DzEq1_P, DzEq2_P], [Dx(1), Dy(1)]);
(%o8) [Dx(1)=0, Dy(1)=1]
```

Per calcular les derivades de segon ordre cal derivar en les equacions que involucren les derivades de primer ordre. Derivant en aquest sistema, s'obté:

```
(%i9) DzzEq1 : diff(-('diff(x,z,1))*z^2-2*x*z+2*y*('diff(y,z,1))+2*x*('diff(x,z,1)), z) = 0;
      DzzEq2 : diff(('diff(x,z,1))*z-x^2*('diff(y,z,1))-2*x*('diff(x,z,1))*y+x, z) = 0;
(%o9)  $-\left(\frac{d^2}{dz^2}x\right)z^2 - 4\left(\frac{d}{dz}x\right)z + 2y\left(\frac{d^2}{dz^2}y\right) + 2\left(\frac{d}{dz}y\right)^2 + 2x\left(\frac{d^2}{dz^2}x\right) + 2\left(\frac{d}{dz}x\right)^2 - 2x = 0$ 
(%o10)  $\left(\frac{d^2}{dz^2}x\right)z - x^2\left(\frac{d^2}{dz^2}y\right) - 4x\left(\frac{d}{dz}x\right)\left(\frac{d}{dz}y\right) - 2x\left(\frac{d^2}{dz^2}x\right)y - 2\left(\frac{d}{dz}x\right)^2y + 2\left(\frac{d}{dz}x\right) = 0$ 
```

Per calcular les derivades de segon ordre $D^2x(1)$, $D^2y(1)$ cal aplicar els valors numèrics corresponents en aquestes equacions:

```
(%i11) DzzEq1_P : -D2x(1)*1 - 4*0*1 + 2*1*D2y(1) + 2*1 + 2*1*D2x(1) + 2*0 - 2*1 = 0;
      DzzEq2_P : D2x(1)*1 - 1*D2y(1) - 4*1*0*1 - 2*1*D2x(1)*1 - 2*0*1 + 2*0 = 0;
(%o11) 2 D2y(1)+D2x(1)=0
(%o12) -D2y(1)-D2x(1)=0
```

Resolent aquest sistema lineal resulta:

```
(%i13) linsolve([DzzEq1_P, DzzEq2_P], [D2x(1), D2y(1)]);  
(%o13) [D2x(1)= 0, D2y(1)= 0]
```

11-3.- Aproximació de funcions diferenciables. Fórmula de Taylor.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_11-3.wxm**.

Definició (polinomi de Taylor d'una funció). Considerem una funció real de diverses variables $f: A \subset \mathbb{R}^n \rightarrow \mathbb{R}$ i $a = (a_1, \dots, a_n)$ un punt interior del domini de la funció. Si la funció és de tipus C^p en aquest punt, s'anomena polinomi de Taylor de la funció en aquest punt, el polinomi de grau p :

$$T_p(x_1, \dots, x_n) = f(a) + \frac{1}{1!} df_a(h) + \frac{1}{2!} d^2 f_a(h^{(2)}) + \dots + \frac{1}{p!} d^p f_a(h^{(p)})$$

$$h = (x_1 - a_1, x_2 - a_2, \dots, x_n - a_n)$$

En aquesta expressió intervé l'anomenada diferencial d'ordre k de la funció, que es defineix mitjançant:

$$d^r f_a(h^{(r)}) = \sum_{i_1, \dots, i_r=1}^n D_{i_1 \dots i_r}^r f(a) dx_{i_1} \cdots dx_{i_r}, \quad h = (dx_1, \dots, dx_n) \in \mathbb{R}^n$$

El programa wxMaxima ens il·lustra com fa els càlculs del polinomi de Taylor d'una funció; de passada anem veient la sintaxi de la instrucció corresponent:

```
(%i7) T[1](x,y) = taylor(f(x,y), [x,y], [x0,y0], [1,1]);
```

$$T_1(x,y) = f(x_0, y_0) + \left(\left. \frac{d}{dx} f(x,y) \right|_{x=x_0} \right) (x-x_0) + \left(\left. \frac{d}{dy} f(x,y) \right|_{y=y_0} \right) (y-y_0) + \dots$$

```
(%i8) T[2](x,y) = taylor(f(x,y), [x,y], [x0,y0], [2,2]);
```

$$T_2(x,y) = f(x_0, y_0) + \left(\left. \frac{d}{dx} f(x,y) \right|_{x=x_0} \right) (x-x_0) + \left(\left. \frac{d}{dy} f(x,y) \right|_{y=y_0} \right) (y-y_0) +$$

$$\frac{\left(\left. \frac{d^2}{dx^2} f(x,y) \right|_{x=x_0} \right) (x-x_0)^2 + 2 \left(\left. \frac{d^2}{dx dy} f(x,y) \right|_{x=x_0} \right) (y-y_0) (x-x_0) + \left(\left. \frac{d^2}{dy^2} f(x,y) \right|_{y=y_0} \right) (y-y_0)^2}{2} +$$

Per tant, observem que la metodologia per al càlcul del polinomi de Taylor de grau n d'una funció diferenciable cal definir la funció, aplicar la instrucció "Taylor()" indicant a quina funció ens referim, les variables, les coordenades del punt i el grau en que cal truncar el desenvolupament, és a dir, el grau del polinomi. Vegem detalls al respecte mitjançant alguns exemples.

Exemple 11.3.1. Es considera la funció diferenciable

$$f(x, y) = \exp(-x + 2y), (x, y) \in \mathbb{R}^2.$$

Es tracta de calcular els polinomis de Taylor de graus 1, 2, 3 d'aquesta funció en el punt $a = (0, 0)$.

En primer lloc definirem la funció:

```
(%i1) f(x,y):=exp(-x+2*y);
(%o1) f(x,y):=exp(-x+2 y)
```

Càlcul del polinomi de Taylor de grau 1:

```
(%i2) T[1](x,y) = taylor(f(x,y), [x,y], [0,0], [1,1]);
(%o2)/T/ T1(x,y)=1+(-x+2 y)+...
```

Càlcul del polinomi de Taylor de grau 2:

```
(%i3) T[2](x,y) = taylor(f(x,y), [x,y], [0,0], [2,2]);
(%o3)/T/ T2(x,y)=1+(-x+2 y)+ $\frac{x^2-4 y x+4 y^2}{2}$ +...
(%i4) ratsimp(%);
(%o4) T2(x,y)= $\frac{4 y^2+(4-4 x) y+x^2-2 x+2}{2}$ 
(%i5) expand(%);
(%o5) T2(x,y)=2 y2-2 x y+2 y+ $\frac{x^2}{2}$ -x+1
```

Càlcul del polinomi de Taylor de grau 3:

```
(%i6) T[3](x,y) = taylor(f(x,y), [x,y], [0,0], [3,3]);
(%o6)/T/ T3(x,y)=1+(-x+2 y)+ $\frac{x^2-4 y x+4 y^2}{2}$ - $\frac{x^3-6 y x^2+12 y^2 x-8 y^3}{6}$ +...
(%i7) ratsimp(%);
(%o7) T3(x,y)= $\frac{8 y^3+(12-12 x) y^2+(6 x^2-12 x+12) y-x^3+3 x^2-6 x+6}{6}$ 
(%i8) expand(%);
(%o8) T3(x,y)= $\frac{4 y^3}{3}$ -2 x y2+2 y2+x2 y-2 x y+2 y- $\frac{x^3}{6}$ + $\frac{x^2}{2}$ -x+1
```

Exemple 11.3.2. Es considera la funció diferenciable

$$f(x, y) = \exp(-x + 2y), (x, y) \in \mathbb{R}^2.$$

Es tracta de calcular els polinomis de Taylor de graus 1, 2, 3 d'aquesta funció en el punt $a = (1, 2)$.

En primer lloc definirem la funció:

```
(%i1) f(x,y):=exp(-x+2*y);
(%o1) f(x,y):=exp(-x+2 y)
```

Càlcul del polinomi de Taylor de grau 1:

```
(%i2) T[1](x,y) = taylor(f(x,y), [x,y], [1,2], [1,1]);
(%o2)/T/ T1(x,y)=%e3+(-%e3(x-1)+2 %e3(y-2))+...
(%i3) ratsimp(%);
(%o3) T1(x,y)=2 %e3 y-%e3 x-2 %e3
```

Càlcul del polinomi de Taylor de grau 2:

```
(%i4) T[2](x,y) = taylor(f(x,y), [x,y], [1,2], [2,2]);
(%o4)/T/ T2(x,y)=%e3+(-%e3(x-1)+2 %e3(y-2))+  


$$\frac{\%e^3 (x-1)^2 - 4 \%e^3 (y-2) (x-1) + 4 \%e^3 (y-2)^2}{2} + \dots$$

(%i5) ratsimp(%);
(%o5) T2(x,y)=
$$\frac{4 \%e^3 y^2 + (-4 \%e^3 x - 8 \%e^3) y + \%e^3 x^2 + 4 \%e^3 x + 5 \%e^3}{2}$$

(%i6) expand(%);
(%o6) T2(x,y)=2 %e3 y2-2 %e3 x y-4 %e3 y+
$$\frac{\%e^3 x^2}{2} + 2 \%e^3 x + \frac{5 \%e^3}{2}$$

```

Càlcul del polinomi de Taylor de grau 3:

```
(%i7) T[3](x,y) = taylor(f(x,y), [x,y], [1,2], [3,3]);
(%o7)/T/ T3(x,y)=%e3+(-%e3(x-1)+2 %e3(y-2))+  


$$\frac{\%e^3 (x-1)^2 - 4 \%e^3 (y-2) (x-1) + 4 \%e^3 (y-2)^2}{2} -$$


$$\frac{\%e^3 (x-1)^3 - 6 \%e^3 (y-2) (x-1)^2 + 12 \%e^3 (y-2)^2 (x-1) - 8 \%e^3 (y-2)^3}{6} + \dots$$

(%i8) ratsimp(%);
(%o8) T3(x,y)=
$$\frac{8 \%e^3 y^3 + (-12 \%e^3 x - 24 \%e^3) y^2 + (6 \%e^3 x^2 + 24 \%e^3 x + 30 \%e^3) y - \%e^3 x^3 - 6 \%e^3 x^2 - 15 \%e^3 x - 12 \%e^3}{6}$$

```

```
(%i9) expand(%);
(%o9) T3(x,y)=
$$\frac{4e^3y^3}{3}-2e^3xy^2-4e^3y^2+e^3x^2y+4e^3xy+5e^3y-\frac{e^3x^3}{6}-e^3x^2-\frac{5e^3x}{2}-2e^3$$

```

Vegem ara un exemple amb una funció de 3 variables.

Exemple 11.3.3. Es considera la funció diferenciable

$$g(x,y,z) = \exp(x+y+2z), \quad (x,y,z) \in \mathbb{R}^3.$$

Es tracta de calcular els polinomis de Taylor de graus 1, 2, 3 d'aquesta funció en el punt $a = (0,0,0)$.

En primer lloc definirem la funció:

```
(%i1) g(x,y,z):=exp(x+y+2*z);
(%o1) g(x,y,z):=exp(x+y+2 z)
```

Càlcul del polinomi de Taylor de grau 1:

```
(%i2) T[1](x,y,z) = taylor(g(x,y,z), [x,y,z], [0,0,0] , [1,1,1]);
(%o2)/T/ T1(x,y,z)=1+(x+y+2 z)+...
```

Càlcul del polinomi de Taylor de grau 2:

```
(%i3) T[2](x,y,z) = taylor(g(x,y,z), [x,y,z], [0,0,0] , [2,2,2]);
(%o3)/T/ T2(x,y,z)=1+(x+y+2 z)+
$$\frac{x^2+(2y+4z)x+y^2+4zy+4z^2}{2}+...$$

(%i4) ratsimp(%);
(%o4) T2(x,y,z)=
$$\frac{4z^2+(4y+4x+4)z+y^2+(2x+2)y+x^2+2xz}{2}$$

(%i5) expand(%);
(%o5) T2(x,y,z)=
$$2z^2+2yz+2xz+2z+\frac{y^2}{2}+xy+y+\frac{x^2}{2}+x+1$$

```

Càlcul del polinomi de Taylor de grau 3:

```
(%i6) T[3](x,y,z) = taylor(g(x,y,z), [x,y,z], [0,0,0] , [3,3,3]);
(%o6)/T/ T3(x,y,z)=1+(x+y+2 z)+
$$\frac{x^2+(2y+4z)x+y^2+4zy+4z^2}{2}+$$


$$\frac{x^3+(3y+6z)x^2+(3y^2+12zy+12z^2)x+y^3+6zy^2+12z^2y+8z^3}{6}+...$$

```



```
(%i7) ratsimp(%);
(%o7) T3(x,y,z)=

$$\frac{8z^3 + (12y + 12x + 12)z^2 + (6y^2 + (12x + 12)y + 6x^2 + 12x + 12)z + y^3 + (3x + 3)y^2 + (3x^2 + 6x + 6)y + x^3 + 3x^2 + 6x}{6}$$

(%i8) expand(%);
(%o8) T3(x,y,z)= $\frac{4z^3}{3} + 2yz^2 + 2xz^2 + 2z^2 + y^2z + 2xyyz + 2yzy + x^2z + 2xzz + 2z + \frac{y^3}{6} + \frac{xy^2}{2} + \frac{y^2}{2} + \frac{x^2y}{2} + xy + y + \frac{x^3}{6} + \frac{x^2}{2} + x + 1$ 
```

11-4.- Extrems de funcions reals de diverses variables.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_11-4.wxm**.

11.4.1 Extrems locals de funcions reals de dues variables

Extrems locals o relatius. Considerem una funció real de dues variables $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$; es diu que la funció té un màxim local o relatiu en el punt $(x_0, y_0) \in A$ si existeix $r > 0$ tal que

$$f(x_0, y_0) \geq f(x, y), \quad \forall (x, y) \in B((x_0, y_0); r) \cap A$$


De manera anàloga, es diu que la funció té un mínim local o relatiu en el punt $(x_0, y_0) \in A$ si existeix $r > 0$ tal que

$$f(x_0, y_0) \leq f(x, y), \quad \forall (x, y) \in B((x_0, y_0); r) \cap A$$

Els màxims i mínims locals es denominen conjuntament extrems locals o relatius de la funció. Vegem alguns exemples amb representació gràfica de wxMaxima.

Mínim local:


```
(%i1) f1(x,y):=x^2+y^2;  
(%o1) f1(x,y):=x^2+y^2  
  
(%i2) wxplot3d(f1(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set hidden3d"])$
```



Màxim local:

```
(%i3) f2(x,y):=1-x^2-y^2;
(%o3) f2(x,y):=1-x^2-y^2

(%i4) wxplot3d(f2(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set hidden3d"])$
```



Condicció necessària d'extrem local. Si una funció $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ té un extrem local en un punt $(x_0, y_0) \in A$ i la funció és diferenciable en aquest punt, aleshores es compleix

$$D_x f(x_0, y_0) = D_y f(x_0, y_0) = 0$$


Aquesta condició significa que la diferencial en el punt és l'aplicació nul·la i que el pla tangent a la superfície $z = f(x, y)$ en el punt $(x_0, y_0, z_0 = f(x_0, y_0))$ és horitzontal.

Cal observar que el recíproc no és cert, és a dir, complint-se la condició de derivades parcials nul·les, pot ser que la funció no tingui cap extrem local en aquest punt, com per exemple, la funció $f(x, y) = x^3 + y^3$, en el punt $(0, 0)$.

Punts crítics d'una funció. S'anomenen punts crítics d'una funció aquells en els quals les derivades parcials de primer ordre són nul·les. El resultat anterior estableix que, per a una funció diferenciable, tot extrem local és un punt crític; així mateix, queda clar que no tot punt crític correspon a un extrem local. Un punt crític en el qual no hi ha extrem local, s'anomena punt de sella. Així mateix pot haver-hi un extrem local en un punt on la funció no sigui diferenciable.


Exemple de punt de sella:

```
(%i5) f3(x,y):=x^3+y^3;  
(%o5) f3(x,y):=x^3+y^3  
  
(%i6) wxplot3d(f3(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set hidden3d"])$
```



Exemple de mínim local que no és punt crític:

```
(%i7) f4(x,y):=+sqrt(x^2 + y^2);  
(%o7) f4(x,y):=sqrt(x^2+y^2)  
  
(%i8) wxplot3d(f4(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set hidden3d"])$
```



Condicions suficients d'extrem local. Considerem $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ diferenciable i suposem que $(x_0, y_0) \in A$ és un punt crític de la funció. El signe de la diferència $f(x, y) - f(x_0, y_0)$ està donat per la forma quadràtica $d^2 f_{(x_0, y_0)}(h^{(2)})$ i es compleix:

- Si $d^2 f_{(x_0, y_0)}(h^{(2)}) \geq 0$ en un entorn del punt $(x_0, y_0) \in A$, aleshores hi haurà un mínim local en aquest punt;
- Si $d^2 f_{(x_0, y_0)}(h^{(2)}) \leq 0$ en un entorn del punt $(x_0, y_0) \in A$, aleshores hi haurà un màxim local en aquest punt;
- Si $d^2 f_{(x_0, y_0)}(h^{(2)})$ no té signe constant en un entorn del punt $(x_0, y_0) \in A$, aleshores no hi ha extrem local (punt de sella).
- Si no es pot saber si $d^2 f_{(x_0, y_0)}(h^{(2)})$ té signe constant en un entorn del punt $(x_0, y_0) \in A$, aleshores no es pot saber si hi ha o no extrem local.

Per saber quan es compleix cada una d'aquestes condicions, recordem que l'expressió de la forma quadràtica $d^2 f_{(x_0, y_0)}(h^{(2)})$ és

$$d^2 f_{(x_0, y_0)}(h^{(2)}) = \begin{pmatrix} x - x_0 & y - y_0 \end{pmatrix} Hf(x_0, y_0) \begin{pmatrix} x - x_0 \\ y - y_0 \end{pmatrix}$$

El signe de la forma quadràtica es pot determinar mitjançant els valors propis de la matriu hessiana $Hf(x_0, y_0)$, que existeixen i són reals, ja que aquesta matriu és simètrica. A més, el signe d'aquests valors propis està determinat pels menors principals de la matriu, de la manera següent. Si es designa:

$$Hf(x_0, y_0) = \begin{pmatrix} D_{xx}f(x_0, y_0) & D_{xy}f(x_0, y_0) \\ D_{xy}f(x_0, y_0) & D_{yy}f(x_0, y_0) \end{pmatrix} = \begin{pmatrix} A & B \\ B & C \end{pmatrix}$$

aleshores:

- Si es compleix $AC - B^2 > 0$, aleshores els valors propis són del mateix signe, verificant-se que:
 - Si $A > 0$ els valors propis són estrictament positius i la forma quadràtica és definida positiva, i en aquest cas l'extrem local és un mínim.
 - Si $A < 0$ els valors propis són estrictament negatius i la forma quadràtica és definida negativa, i en aquest cas l'extrem local és un màxim.
- Si es compleix $AC - B^2 < 0$, aleshores els valors propis són de signe diferent, i en aquest cas la forma quadràtica és indefinida i no hi ha extrem local, és a dir, es tracta d'un punt de sella.
- Si es compleix $AC - B^2 = 0$, aleshores algun dels valors propis és zero i el criteri no permet decidir sobre la naturalesa del punt crític.

Exemple 11.4.1. Es considera la funció diferenciable $f(x, y) = x^3 + 3xy^2 - 15x - 12y$. Es tracta d'estudiar els extrems locals d'aquesta funció.

Definim la funció amb wxMaxima i calculem les derivades parcials de primer ordre:

```
(%i1) f(x,y):=x^3+3*x*y^2-15*x-12*y;
(%o1) f(x,y):=x^3+3 x y^2+(-15)x+(-12)y

(%i2) diff(f(x,y),x); diff(f(x,y), y);
(%o2) 3 y^2+3 x^2-15
(%o3) 6 x y-12

(%i4) Dxf(x,y):=3*y^2+3*x^2-15;    Dyf(x,y):=6*x*y-12;
(%o4) Dxf(x,y):=3 y^2+3 x^2-15
(%o5) Dyf(x,y):=6 x y-12
```

Escrivim les equacions que resulten d'igualar a zero les derivades parcials i resollem el sistema d'equacions:

```
(%i6) Eq1:Dxf(x,y)=0;    Eq2:Dyf(x,y)=0;    algsys([Eq1, Eq2], [x,y]);
(%o6) 3 y^2+3 x^2-15=0
(%o7) 6 x y-12=0
(%o8) [[x=2,y=1],[x=1,y=2],[x=-1,y=-2],[x=-2,y=-1]]
```

Les solucions d'aquest sistema són:

$$P_1(2,1); P_2(1,2); P_3(-1,-2); P_4(-2,-1)$$

Càlcul de la matriu hessiana en un punt (x,y) :

```
(%i9) 'Hf(x,y) = Hf: hessian(f(x,y), [x,y]);
(%o9) Hf(x,y)=

$$\begin{bmatrix} 6x & 6y \\ 6y & 6x \end{bmatrix}$$

```

Estudi de cada un dels punts crítics.

Punt $P_1(2,1)$:

```
(%i10) 'Hf(2,1)= Hf_P1: subst([x=2,y=1],Hf);
(%o10) Hf(2,1)=

$$\begin{bmatrix} 12 & 6 \\ 6 & 12 \end{bmatrix}$$

```

```
(%i11) determinant(Hf_P1);  
(%o11) 108
```

Per tant la funció té un mínim local en P1(2,1).

Punt P2(1,2):

```
(%i12) 'Hf(1,2)= Hf_P2: subst([x=1,y=2],Hf);  
(%o12) Hf(1,2)=
$$\begin{bmatrix} 6 & 12 \\ 12 & 6 \end{bmatrix}$$

```

```
(%i13) determinant(Hf_P2);  
(%o13) -108
```

Per tant la funció té un punt de sella en P2(1,2).

Punt P3(-1, -2):

```
(%i14) 'Hf(-1,-2)= Hf_P3: subst([x=-1,y=-2],Hf);  
(%o14) Hf(-1,-2)=
$$\begin{bmatrix} -6 & -12 \\ -12 & -6 \end{bmatrix}$$

```

```
(%i15) determinant(Hf_P3);  
(%o15) -108
```

Per tant la funció té un punt de sella en P3(-1, -2).

Punt P4(-2, -1):

```
(%i16) 'Hf(-2,-1)= Hf_P4: subst([x=-2,y=-1],Hf);  
(%o16) Hf(-2,-1)=
$$\begin{bmatrix} -12 & -6 \\ -6 & -12 \end{bmatrix}$$

```

```
(%i17) determinant(Hf_P4);  
(%o17) 108
```

Per tant la funció té un màxim local en el punt P4(-2, -1).

Exemple 11.4.2. Es considera la funció diferenciable $f(x, y) = (x + y)^2$. Es tracta d'estudiar els extrems locals d'aquesta funció.

Definim la funció amb wxMaxima i calculem les derivades parcials de primer ordre:

```
(%i1) f(x,y):=(x+y)^2; diff(f(x,y),x); diff(f(x,y), y);
(%o1) f(x,y):=(x+y)^2
(%o2) 2(y+x)
(%o3) 2(y+x)

(%i4) Dxf(x,y):=2*(x+y); Dyf(x,y):=2*(x+y);
(%o4) Dxf(x,y):=2(x+y)
(%o5) Dyf(x,y):=2(x+y)
```

Escrivim les equacions que resulten d'igualar a zero les derivades parcials i resollem el sistema d'equacions:

```
(%i6) Eq1:Dxf(x,y)=0; Eq2:Dyf(x,y)=0;
      algsys([Eq1, Eq2], [x,y]);
(%o6) 2(y+x)=0
(%o7) 2(y+x)=0
(%o8) [[x=%r7, y=-%r7]]
```


La solució d'aquest sistema és $P(x_0, -x_0)$. Càlcul de la matriu hessiana en un punt (x, y) :

```
(%i9) 'Hf(x,y)=Hfxy:hessian(f(x,y), [x,y]);
(%o9) Hf(x,y)= $\begin{bmatrix} 2 & 2 \\ 2 & 2 \end{bmatrix}$ 

(%i10) determinant(Hfxy);
(%o10) 0
```

Per tant, el criteri no permet decidir sobre la naturalesa dels punts crítics de la funció. Tanmateix, observi's que es compleix $f(x, y) \geq 0$, $\forall (x, y) \in \mathbb{R}^2$ i que en els punts $P(x_0, -x_0)$ la funció compleix $f(x_0, -x_0) = 0$ i, per tant, en aquests punts la funció assolix el seu mínim absolut, tal com es pot veure a la representació gràfica de la funció.


```
(%i11) wxplot3d(f(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set hidden3d"])$
```



11.4.2 Extrems condicionats. El mètode de Lagrange.

Considerem una funció real de dues variables $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$; si es considera un subconjunt $U \subset A$ definit per $U = \{(x, y) : g(x, y) = 0\}$ els extrems de la restricció de f al subconjunt U s'anomenen extrems condicionats o l·ligats de f . La funció f s'acostuma a denominar funció objectiu.

El mètode de Lagrange consisteix en considerar l'anomenada funció de Lagrange

$$L(x, y, \lambda) = f(x, y) + \lambda g(x, y)$$

i resoldre el sistema homogeni format per les 3 equacions següents: les derivades parcials de primer ordre de la funció igualades a zero (2 equacions) i l'equació definida per la condició, és a dir, determinar les solucions del sistema:

$$\begin{cases} D_x L(x, y, \lambda) = D_x f(x, y) + \lambda D_x g(x, y) = 0 \\ D_y L(x, y, \lambda) = D_y f(x, y) + \lambda D_y g(x, y) = 0 \\ g(x, y) = 0 \end{cases}$$

Aquest mètode s'aplica per calcular els extrems absoluts d'una funció, concretament quan la condició expressa la pertinença del punt a la frontera del conjunt.

11.4.3 Extrems absoluts d'una funció sobre un compacte.

Considerem una funció real de dues variables $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$; es diu que la funció té un màxim absolut en el punt $(x_0, y_0) \in A$ si es compleix

$$f(x_0, y_0) \geq f(x, y), \quad \forall (x, y) \in A$$

De manera anàloga, es diu que la funció té un mínim absolut en el punt $(x_0, y_0) \in A$ si es compleix

$$f(x_0, y_0) \leq f(x, y), \quad \forall (x, y) \in A$$

Els màxims i mínims absoluts es denominen conjuntament extrems absoluts de la funció.

Càlcul d'extrems absoluts d'una funció. Si una funció $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ és contínua en el seu camp d'existència A , el teorema de Weierstrass estableix que la funció assoleix extrems absoluts en qualsevol subconjunt $W \subset A$ si aquest subconjunt és tancat i acotat (és a dir, compacte). Els extrems absoluts d'una funció en un subconjunt compacte W s'assoleixen en alguna de les situacions següents:

- Punts crítics de l'interior de W , en els quals la funció sigui diferenciable;
- Punts de l'interior de W , en els quals la funció no sigui diferenciable.
- Punts de la frontera de W ;
- Punts de la frontera de W en els quals no hi hagi recta tangent.

Per tant, per tal de determinar els extrems absoluts d'una funció en un subconjunt compacte cal determinar en cada un dels casos els punts en els quals la funció pot tenir un extrem: en el primer cas, es calculen els extrems locals; en el segon cas cal determinar els punts en que la funció no sigui diferenciable i fer una anàlisi local; en el tercer cal aplicar el mètode de Lagrange amb la condició de pertànyer a la frontera i, finalment, estudiar si és el cas, els punts en que la frontera no té recta tangent (frontera no derivable).

Exemple 11.4.3. Es tracta de calcular els extrems absoluts de la funció

$$f(x, y) = 2(x-1)^2 + 2(y-1)^2, \quad (x, y) \in \mathbb{R}^2$$

en el conjunt que s'indica:

$$W = \{(x, y) : x^2 + y^2 - 9 \leq 0, xy \geq 0\}$$

A la Figura 11.4.1 es representa esquemàticament aquest conjunt.




Figura 11.4.1

S'observa que el domini de la funció és $\mathbb{R}^2$ i que la funció és diferenciable, i per tant és contínua, en el seu domini. El subconjunt W és tancat i acotat, és a dir compacte i, per tant, la funció té extrems absoluts en aquest subconjunt.

Definim la funció i calculem els punts crítics de la funció en l'interior de W :

```
(%i1) f(x,y):=2*(x-1)^2+2*(y-1)^2;
(%o1) f(x,y):=2(x-1)^2+2(y-1)^2

(%i2) Dxf=diff(f(x,y), x); Dyf=diff(f(x,y), y);
(%o2) Dxf = 4(x-1)
(%o3) Dyf = 4(y-1)
```

L'únic punt crític és $P1(1,1)$.

La frontera de W és el conjunt:

$$\partial W = \{x^2 + y^2 = 9, xy \geq 0\} \cup \{x = 0, -3 \leq y \leq 3\} \cup \{y = 0, -3 \leq x \leq 3\}.$$

Caldrà estudiar la funció de Lagrange en cada un dels tres subconjunts i en els punts en els quals la frontera no és derivable: $(0,0)$, $(-3,0)$, $(3,0)$, $(0,-3)$ i $(0,3)$.

Calculem la funció de Lagrange en el primer conjunt i plantegem i resollem el sistema d'equacions que resulten d'igualar les derivades parcials a zero:

```
(%i5) L1(x,y,a):=f(x,y)+a*(x^2+y^2-9);
      Eq1a:diff(L1(x,y,a), x)=0;      Eq1b:diff(L1(x,y,a), y)=0;      Eq1c:diff(L1(x,y,a), a)=0;
      algsys([Eq1a, Eq1b, Eq1c], [x,y,a]);
(%o5) L1(x,y,a):=f(x,y)+a(x^2+y^2-9)
(%o6) 2 a x + 4 (x - 1) = 0
(%o7) 2 a y + 4 (y - 1) = 0
(%o8) y^2 + x^2 - 9 = 0
(%o9) [[x =  $\frac{3}{\sqrt{2}}$ , y =  $\frac{3}{\sqrt{2}}$ , a =  $\frac{2^{3/2}-6}{3}$ ], [x =  $-\frac{3}{\sqrt{2}}$ , y =  $-\frac{3}{\sqrt{2}}$ , a =  $-\frac{2^{3/2}+6}{3}$ ]]
```

Solucions del sistema: $P2\left(\frac{3}{\sqrt{2}}, \frac{3}{\sqrt{2}}\right)$ i $P3\left(-\frac{3}{\sqrt{2}}, -\frac{3}{\sqrt{2}}\right)$.

Calculem la funció de Lagrange en el segon conjunt i plantegem i resollem el sistema d'equacions que resulten d'igualar les derivades parcials a zero:

```
(%i12) L2(x,y,b):=f(x,y)+b*x;
      Eq2a:diff(L2(x,y,b), x)=0;      Eq2b:diff(L2(x,y,b), y)=0;      Eq2c:diff(L2(x,y,b), b)=0;
      algsys([Eq2a, Eq2b, Eq2c], [x,y,b]);
(%o12) L2(x,y,b):=f(x,y)+b x
(%o13) 4 (x - 1) + b = 0
(%o14) 4 (y - 1) = 0
(%o15) x = 0
(%o16) [[x = 0, y = 1, b = 4]]
```

Solució del sistema: $P4(0,1)$.

Calculem la funció de Lagrange en el tercer conjunt i plantegem i resollem el sistema d'equacions que resulten d'igualar les derivades parcials a zero:

```
(%i18) L3(x,y,c):=f(x,y)+c*y;
      Eq3a:diff(L3(x,y,c), x)=0;      Eq3b:diff(L3(x,y,c), y)=0;      Eq3c:diff(L3(x,y,c), c)=0;
      algsys([Eq3a, Eq3b, Eq3c], [x,y,c]);
(%o18) L3(x,y,c):=f(x,y)+c y
(%o19) 4 (x - 1) = 0
(%o20) 4 (y - 1) + c = 0
(%o21) y = 0
(%o22) [[x = 1, y = 0, c = 4]]
```

Solució del sistema: $P5(1,0)$.

Ara calculem el valor de la funció en cada un dels punts determinats:

```
(%i24) 'f(1,1)=f(1,1);
(%o24) f(1,1)=0
```

```

(%i25) 'f(P2)=f(3/sqrt(2),3/sqrt(2)), numer;
      'f(P3)=f(-3/sqrt(2),-3/sqrt(2)), numer;
(%o25) f(P2)= 5.029437251522857
      f(P3)= 38.97056274847714
(%i27) 'f(P4)=f(0,1);   'f(P5)=f(1,0);
(%o27) f(P4)= 2
      f(P5)= 2
(%i29) 'f(P6)=f(0,0);   'f(P7)=f(3,0);   'f(P8)=f(0,3);
      'f(P9)=f(-3,0);   'f(P10)=f(0,-3);
(%o29) f(P6)= 4
      f(P7)= 10
      f(P8)= 10
      f(P9)= 34
      f(P10)= 34

```

A la Figura 11.4.2 es representen els punts en els quals es calcula el valor de la funció per tal de determinar els extrems absoluts.




Figura 11.4.2

Finalment calculem el mínim i el màxim d'aquest conjunt de valors:

```

(%i34) min(f(1,1), f(3/sqrt(2),3/sqrt(2)), f(-3/sqrt(2),-3/sqrt(2)),
      f(0,1), f(1,0), f(3,0), f(0,3), f(0,0), f(-3,0), f(0,-3) ), numer;
(%o34) 0

```

```
(%i35) max(f(1,1), f(3/sqrt(2),3/sqrt(2)), f(-3/sqrt(2),-3/sqrt(2)),
          f(0,1), f(1,0), f(3,0), f(0,3), f(0,0), f(-3,0), f(0,-3) ), numer;
(%o35) 38.97056274847714
```

Per tant el mínim absolut és al punt $P1(1,1)$ i el màxim absolut al punt $P3\left(-\frac{3}{\sqrt{2}}, -\frac{3}{\sqrt{2}}\right)$.